Cercles de Ford

Voici une procédure [1] calculant les termes de la suite de Farey à l'ordre n :

```
▷ Farey(n) := block([L:[[0,1]],a:0,b:1,c:1,d:n],
      while 2*c<d do (
          k:floor((b+n)/d),
          e:k*c-a,
          f:k*d-b,
          a:c,b:d,c:e,d:f,
          L:cons([a,b],L)
      ),
      append(reverse(L),[[1,2]],map(lambda([x],[x[2]-x[1],x[2]]),L))
  );
```

```
▷ Farey(5);
```

[[0,1],[1,5],[1,4],[1,3],[2,5],[1,2],[3,5],[2,3],[3,4],[4,5],[1,1]]

Ou encore :

```
▷ map(lambda([x],x[1]/x[2]),Farey(7));
```

$$\left[0,\frac{1}{7},\frac{1}{6},\frac{1}{5},\frac{1}{4},\frac{2}{7},\frac{1}{3},\frac{2}{5},\frac{3}{7},\frac{1}{2},\frac{4}{7},\frac{3}{5},\frac{2}{3},\frac{5}{7},\frac{3}{4},\frac{4}{5},\frac{5}{6},\frac{6}{7},1\right]$$

Définissons le cercle de Ford associé à la fraction irréductible $\frac{p}{q}$.

```
▷ CF(f) := Cercle(Point(f[1]/f[2],1/2/f[2]^2),1/2/f[2]^2);
```

Tout est prêt pour une première figure.

```
▷ Cercles:map(lambda([f],CF(f)),Farey(9))$
```

```
▷ Figure('Cercles);
```



Pour y mettre un peu de couleur, nous allons séparer les cercles suivant leur rayon.

1. À retrouver dans un fil de discussion de la liste maxima.

```
▷ CFL(n) := block([o:[],l,p,q],
      for i:1 thru n do o:cons([],o),
      l:Farey(n),
      for i:1 thru length(l) do (
          [p,q]:l[i],
          o[q]:cons(CF([p,q]),o[q])
      ),
      o
  );
```

```
▷ [C1,C2,C3,C4,C5,C6,C7,C8]:CFL(8)$
```

```
▷ Figure('C1,'C2,'C3,'C4,'C5,'C6,'C7,'C8);
```



pmaxima

```
Maxima 5.30.0 http://maxima.sourceforge.net
using Lisp GNU Common Lisp (GCL) GCL 2.6.7 (a.k.a. GCL)
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
```

Vous trouverez le *style* attaché aux figures à la fin du source de ce document (18 juin 2013).